2. 
[image: image1.wmf],

)

1

(

)

(

100

)

,

(

2

2

2

x

y

x

y

x

z

-

+

-

=



[image: image2.wmf]048

.

2

048

.

2

,

048

.

2

048

.

2

£

£

-

£

£

-

y

x


Рассмотреть одноточечное скрещивание и инверсионную мутацию.

Каждая переменная кодируется 30 битами.

Провести расчеты для 30 и 100 поколений.

Сравнить получающиеся решения при размерах популяции 10, 20, 30 особей.

ЗАДАНИЕ:

необходимо найти минимум функции 

 в заданной области.

Каждый запуск программы необходимо повторять, по крайней мере, 20(30 раз. После этого из набора полученных решений надо отобрать лучшее. Разумеется, это надо сделать, поместив содержание главной программы в соответствующий цикл, в котором будет одновременно выбираться наилучшее решение. Одновременно надо вычислить и среднее значение минимума за эти 20-30 прогонов.

РЕШЕНИЕ:

Так как число переменных в уравнении равно двум, то размерность пространства поиска  dim = 2; 

1. Составим программу:
program popul;

uses crt;

const

  maxpop = 200;

  maxstring = 30;

  dim = 2; {размерность пространства поиска}

type

  allele = boolean;

  {Алель - позиция в битовой строке }

  chromosome = array[1..maxstring*dim] of allele;

  {битовая строка }

  fenotype = array[1..dim] of real;

individual = record

  chrom:chromosome;

  { генотип = битовая строка }

  xy:fenotype; { фенотип = массив вещественных

  координат точки в пространстве поиска }

  fitness:real; {значение целевой функции }

end;

population = array[1..maxpop] of individual;

const

 xymax:real=2.048;

  {массив максимальных значений для координат точки в пространстве поиска}

 xymin:Real=-2.048;

  {массив минимальных значений для координат точки в пространстве поиска}

var

oldpop, newpop, intpop :population;

{Три непересекающихся популяции - старая, новая и промежуточная }

popsize, lchrom, gen, maxgen: integer;

{Глобальные целые переменные}

pcross, pmutation, sumfitness:real;

{глобальные вещественные переменные }

nmutation, ncross:integer;

{Статистические целые }

avg, max, min:real; { Статистические вещественные }

{Вероятностные процедуры }

function random_:real;

begin

  random_ := random(65535)/(65535-1);

end;

function flip(probability:real):boolean;

{подбрасывание монетки - true если орел}

begin

  if probability = 1.0 then

                       flip := true

                      else

                        flip := (random_ <= probability);

end;

function rnd(low,high:integer):integer;

{Случайный выбор между low и high }

var

i:integer;

begin

  if low >= high then i := low

                 else begin

                        i := trunc( random_ * (high-low+1) + low);

                        if i > high then i := high;

                      end;

  rnd := i;

end;

{интерфейсные процедуры: decode and objfunc }

function objfunc(xy:fenotype):real;

begin

{  objfunc:= int(xy[1])+int(xy[2]);}

{ objfunc:=xy[1]*xy[1];}

 objfunc:=100*sqr(xy[1]*xy[1]-xy[2])+sqr(1-xy[1]);

end;

procedure decode(chrom:chromosome; lbits:integer; var xy:fenotype);

{Декодирование строки в массив вещественных координат точки в пространстве поиска - true=1, false=0 }

var

  i,j:integer;

  f, accum, powerof2:real;

begin

  for i:=1 to dim do begin

    accum := 0.0;

    powerof2 := 1;

    f:=1;

    for j := 1+lbits*(i-1) to lbits+lbits*(i-1) do begin

      if chrom[j] then accum := accum + powerof2;

      powerof2 := powerof2 * 2;

      f:=f*2;

    end;

    xy[i] := xymin+(xymax-xymin)*accum/(f-1)

  end;

end;

{Расчет статистических величин: statistics }

procedure statistics(popsize:integer; var max,avg,min,sumfitness:real;var pop:population);

{Расчет статистик популяции }

var j:integer;

begin

  {Инициализация }

  sumfitness := pop[1].fitness;

  min := pop[1].fitness;

  max := pop[1].fitness;

  {Цикл для max, min, sumfitness }

  for j := 2 to popsize do with pop[j] do begin

    sumfitness := sumfitness + fitness;

    {Накопление суммы значений функции пригодности }

    if fitness>max then max := fitness;

    { Новое значение max }

    if fitness<min then min := fitness;

    { Новое значение min }

  end;

  {Расчет среднего }

  avg := sumfitness/popsize;

end;

{Процедура инициализации initpop}

procedure initpop;

{Инициализация начальной популяции случайным образом }

var

 j, j1:integer;

begin

  for j := 1 to popsize do with oldpop[j] do begin

    for j1 := 1 to lchrom*dim do chrom[j1] := flip(0.5);

    {Бросок монетки}

    decode(chrom,lchrom,xy);

    { Декодирование строки }

    fitness := objfunc(xy);

    { Вычисление начальных значений функции пригодности }

  end;

end;

{ 3 генетических оператора: отбора (select), скрещивания (crossover),

и мутации (mutation) }

procedure select;

    {процедура выбора}

var

  ipick:integer;

  procedure shuffle(var pop:population);

  {процедура перемешивания популяции в процессе отбора}

  var

  i,j:integer;

  ind0:individual;

  begin

    for i := popsize downto 2 do begin

      j:= random(i-1)+1;

      ind0:=pop[i];

      pop[i]:=pop[j];

      pop[j]:=ind0;

    end;

  end;

  function select_1:integer;

  var

   j1,j2,m:integer;

  begin

    if (ipick>popsize) then begin

      shuffle(oldpop);

      ipick:=1

    end;

    j1:=ipick;

    j2:=ipick+1;

    if (oldpop[j2].fitness<oldpop[j1].fitness) then

      m:=j2

    else

      m:=j1;

    ipick:=ipick+2;

    select_1:=m;

  end;

var

  j:integer;

begin

  ipick:=1;

  for j:=1 to popsize do begin

    intpop[j]:=oldpop[select_1];

  end;

  oldpop:=intpop;

end;

procedure mutation(pmutation: real; var nmutation:integer;var mchrom:chromosome);

{мутация Двухточечная с вероятностью pmutation}

var

  mutate:boolean;

  i: integer;

  aa: allele;

begin

  mutate := flip(pmutation); {выбирается вероятность мутации (1%)}

  { Flip the biased coin }

  if mutate then begin

     for i:=1 to maxstring*dim do begin

      if (mchrom[i]=TRUE) then mchrom[i]:=FALSE else mchrom[i]:=TRUE;

     end;

  end else;

  { нет мутации }

end;

procedure crossover(var parent1, parent2, child1, child2:chromosome;

      flchrom:integer; var ncross, nmutation, jcross:integer;

var pcross, pmutation:real);

{Скрещивание 2 родительских строк, результат помещается в 2 строках-потомках }

var

j:integer;

begin

  if flip(pcross) then begin

    { Выполняется скрещивание с вероятностью pcross }

    jcross := rnd(1,flchrom-1);

    { Определение точки сечения в диапазоне между 1 и l-1 }

    ncross := ncross + 1;

    {Инкрементирование счетчика скрещиваний }

  end else

    jcross := flchrom;

  {певая часть обмена обмена, 1 to 1 and 2 to 2 }

  for j := 1 to jcross do begin

    child1[j] := parent1[j];

    child2[j] := parent2[j];

  end;

  {вторая часть обмена, 1 to 2 and 2 to 1 ]}

  if jcross<>flchrom then

  { пропуск, если точка скрещивания равна flchrom--скрещивание не происходит }

  for j := jcross+1 to flchrom do begin

    child1[j] := parent2[j];

    child2[j] := parent1[j];

  end;

  mutation(pmutation,nmutation,child1);

  mutation(pmutation,nmutation,child2);

end;

{ Процедура создания нового поколения: generation }

procedure generation;

{Генерирование нового поколения при помощи отбора, скрещивания и мутации }

{Прим: предполагается, что популяция имеет четный размер}

var

j, mate1, mate2, jcross:integer;

begin

  select;

  j := 1;

  repeat

    {выполняются отбор, скрещивание и мутация пока не полностью не сформируется новая популяция newpop }

    mate1 := j;

    { выбор родительской пары }

    mate2 := j+1;

    {скрещивание и мутация - мутация вставлена в процедуру скрещивания }

    crossover(oldpop[mate1].chrom, oldpop[mate2].chrom,

    newpop[j ].chrom, newpop[j + 1].chrom,

    lchrom*dim, ncross, nmutation, jcross, pcross, pmutation);

    {Декодирование строки и вычисление пригодности }

    with newpop[j ] do begin

      decode(chrom, lchrom,xy);

      fitness := objfunc(xy);

    end;

    with newpop[j+1] do begin

      decode(chrom, lchrom,xy);

      fitness := objfunc(xy);

    end;

    j := j + 2;

  until j>popsize

end;

BEGIN { Главная программа }

  popsize:=30;              {10,20,30}

  {размер популяции}

  lchrom:=30;

  {число битов на одиn кодируемый параметр}

  maxgen:=100;              {30 & 100}

  {максимальное число поколений}

  pmutation:=0.01;

  {вероятность мутации}

  {вероятность скрещивания}

  pcross:=0.9;

  { Инициализация генератора случайных чисел }

  randomize;

  clrscr;

  { Инициализация счетчиков }

  nmutation := 0;

  ncross := 0;

  initpop;

  statistics(popsize, max, avg, min, sumfitness, oldpop);

  gen := 0; { Установка счетчика поколений в 0 }

  repeat { Главный итерационный цикл }

    gen := gen + 1;

    generation;

    statistics(popsize, max, avg, min, sumfitness, newpop);

    oldpop := newpop;

    {переход на новое поколение }

    writeln('     min= ',min:5:10);

    delay(100);

  until (gen >= maxgen);

  write('жми ENTER!!!');

  readln;

END. { End главной программы }

2. Проведём серию испытаний. Согласно заданию изменяем значения размера популяции – Popsize: 10, 20, 30 и Maxgen – число поколений:  30, 100. Количество прогонов пример равным 20.
Таблица с данными

Рассчитаем среднее  значение минимума за эти 20:
Zmin(10,30)=
Zmin(20,30)=

Zmin(30,30)=

Zmin(10,100)=
Zmin(20,100)=

Zmin(30,100)=

3. Исследуем заданную функцию в системе MathCad 2000 – построим график её функции.

рис. 1. График функции z(x,y)

Найдём минимум функции. Из графика видно, что минимум функции находится в точке  x=1, y=1 и он равен 0.
4. Вывод. 
Как видно из серии испытаний наибольший размер популяции  даёт более точный результат. При этом число поколений сказывается на точности работы программы, но незначительно. Самый точный результат был получен при размере популяции = 30 и числе поколений = 100. Погрешность составила 2%. Возможно, что полученные в серии испытаний результаты были неточны вследствие сложного вида графика функции.  

